

BASOTTER

ESTENSIONE BASIC
PER IL PLOTTER 1520

COD. 6023

 **commodore**
COMPUTER

BASOTTER

ESTENSIONE BASIC
PER IL PLOTTER 1520

COD. 6023

 **commodore**
COMPUTER

INDICE

PREMESSA	5
INTRODUZIONE	7
INSTALLAZIONE	7
IL FUNZIONAMENTO DEL 1520	9
I COMANDI GRAFICI	9
PENNA	10
LINEA	10
VAI	10
RETTANGOLO	10
TRAT	11
ARCO	11
SETTORE	11
ZERO	12
REL	12
ASS	12
ZERO=	12
SCALA	12
COMANDI DI GESTIONE STRINGHE	15
STAMPA	15
MISURA	15
PLIST	16
INCLINAZIONE	16
MAIUSCOLE-MINUSCOLE	16
I COMANDI "BASIC"	17
ELSE	17
CGOTO-CGOUB	17
CPOKE	18

CPEEK	18
JOY	18
PADDLE	19
HEX\$	19
BIN\$	19
I COMANDI DI GESTIONE DISCO	20
ELENCA	20
CANCELLA	20
CONVALIDA	21
RINOMINA	21
DISTRUGGI	21
INIZIALIZZA	21
DISCO	22
COMANDO "J"	22
COMANDI BASIC "BASOTTER 84"	23
FUNZIONI	24
AQUILA	25
DIAGRAMMI	26
SIN X + SIN X/2	28

PREMESSA

Il BASOTTER è un insieme di programmi realizzati per esaltare e facilitare l'uso del Plotter/stampante 1520.

Questo prodotto è stato realizzato tenendo presente le esigenze dell'utente che, acquistando un prodotto come il Plotter 1520, sia in grado entro breve termine, di usare la macchina in modo produttivo.

Infatti l'implementazione dei comandi basic e la struttura degli stessi, permettono a chiunque, anche digiuno di tecnica di programmazione, di usare il Plotter/stampante 1520 immediatamente.

Ci congratuliamo quindi per l'acquisto da lei fatto e sicuramente converrà con noi che raramente si è avuta la disponibilità di un prodotto che comprenda tante caratteristiche contemporaneamente come: facilità d'uso, basso costo, prestazioni sofisticate e l'ormai famosa affidabilità Commodore.

Grazie per aver scelto Commodore e buon lavoro.

L. Angeli

Commodore Italiana

INTRODUZIONE

Il Basotter è un programma per il Commodore 64 che rende più facile ed immediato l'uso del Plotter 1520.

La configurazione necessaria per l'uso del Basotter è costituita dal Commodore 64, dalla unità a disco 1541, e dal plotter 1520.

INSTALLAZIONE

Collegate l'unità a disco al C64, ed il Plotter all'unità a disco, tramite le porte seriali. Date alimentazione, nell'ordine, a Plotter, disco e C64.

Inserite il dischetto contenente il Basotter nell'unità a disco.

Scrivete il comando `LOAD"*",8,1`.

Aspettate che il C64 carichi il programma dal disco (durante il periodo di caricamento apparirà una figura sullo schermo).

Quando il caricamento è terminato, compare sullo schermo il solito prompt "READY." del Basic.

A questo punto, siete pronti per incominciare; buon divertimento con il Basotter!!!

II FUNZIONAMENTO DEL 1520

Quando il 1520 lavora il modo grafico, le quattro penne di cui dispone possono spaziare in un campo largo ben 480 punti e alto 1999; il 1520 può tracciare linee, spostare la penna da un punto all'altro, cambiare colore e usare un sistema di riferimento definito dall'utente. Tuttavia la gestione di questi comandi basic risultava abbastanza noiosa, a causa dei più files che era necessario tenere aperti, delle sintassi poco naturali, e della difficile gestione basata sul "secondary address". Sotto questo punto di vista, Basotter compie un notevole passo avanti, introducendo comandi dall'uso semplice ed intuitivo, consentendo inoltre di tracciare cerchi, parti di cerchi e settori, di realizzare disegni e diagrammi a torta, nonché disegnare quadrati e rettangoli.

Una delle peculiarità del 1520 è quella di poter lavorare o con il sistema di riferimento standard o con uno definibile. Il sistema standard, attivato all'accensione, consiste nel definire lo spazio entro al quale si muovono le penne come i settori 1 e 4 di un normale sistema di assi cartesiani, le cui ascisse saranno sempre positive. Per lavorare anche con coordinate di ascissa negativa, è necessario ridefinire l'origine del nostro sistema cartesiano, traslando gli assi.

Ad esempio, traslando l'origine nel punto 220,10 (relativo al sistema assoluto precedente), tutti i punti e le linee che verranno disegnate saranno spostate a destra di 220 punti e in alto di 10, rispetto al sistema assoluto normale.

Ciò è molto utile in quanto, oltre a permettere di traslare qualsiasi figura senza apportare grosse modifiche ai programmi, consente di lavorare anche con i settori 2 e 3 del pianto cartesiano e, quindi, di lavorare con punti di ascissa negativa.

I COMANDI GRAFICI

Il Basotter fornisce i seguenti comandi orientati alla gestione della grafica del 1520:

Versione italiana

PENNA

ZERO=

ZERO

LINEA

REL

ASS

VAI

TRAT

ARCO

SETTORE

SCALA

RETTANGOLO

Versione inglese

PEN IS

SET HOME

HOME

LINE

RELATIVE

ASSOLUTE

MOVE

DASH

ARC

SECTOR

SCALE

BOX

PENNA

Il comando penna serve per cambiare il colore con il quale si vuole scrivere. Deve essere seguito dal numero del colore che si vuole selezionare oppure dal nome del colore stesso.

I codici colore sono i seguenti:

0...NERO (BLACK)

1...BLU (BLUE)

2...VERDE (GREEN)

3...ROSSO (RED)

Esempio: battere il comando diretto: PENNA 0 oppure PENNA BLU e premere RETURN.

Il 1520 risponderà facendo ruotare il tamburo portapenne finchè la penna blu non si troverà in cima. Nel caso in cui fosse già selezionata la penna blu il 1520 ignorerà il comando.

LINEA

Il comando linea traccia un segmento. I punti da unire devono essere specificati facendo seguire al comando la ascissa del primo punto, una virgola, l'ordinata del primo punto, un TO, l'ascissa del secondo punto, una virgola e l'ordinata del secondo punto.

LINEA 0,0 TO 100,10

Il 1520 porterà la penna alle coordinate 0,0, e tratterà una linea fino al punto 100,10.

È possibile tracciare più linee con un comando solo, continuando a scrivere TO seguito dalla virgola e dalle coordinate del punto successivo.

Esempio:

LINEA TO 100,100

VAI

VAI serve a portare la penna in un determinato punto. Per portare la penna sul punto 100,200, digitare:

VAI 100,200

La penna si sposterà alla posizione 100,200.

RETTANGOLO

RETTANGOLO disegna quadrati o rettangoli descritti con le coordinate di due spigoli opposti.

La sintassi è simile a quella del comando LINEA, ma non sono accettati più di due punti.

Esempio:

RETTANGOLO 0,0 TO 100,200

disegna un rettangolo la cui diagonale ha per estremi i punti 0,0 e 100,200.

TRAT

TRAT abilita/disabilita il tratteggio automatico. Deve essere seguito da un numero compreso tra 0 e 15. 0 sta ad indicare tratteggio disabilitato, ossia linea continua, 1-15 attivano il tratteggio; 1 seleziona un tratteggio molto fine, 15 molto largo.

Esempio:

TRAT3:RETTANGOLO 100,10 TO 300,400

LINEA 100,10 TO 300,400

TRAT0:RETTANGOLO 0,0 TO 400,200

ARCO

ARCO traccia parti di circonferenza o circonferenze di centro, raggio, e limiti assegnati; la sintassi è la seguente:

ARCO X,Y,R,RAD1,RAD2

dove X è l'ascissa del centro, Y l'ordinata del centro, R il raggio, RAD1 è l'angolo in cui inizia l'arco e RAD2 l'angolo in cui finisce. Ricordarsi che sia RAD1 che RAD2 sono espressi in radianti, e che quindi, la circonferenza completa ha limiti da 0 a $\pi \times 2$.

RAD2e/o RAD1 possono essere omessi, in tal caso il loro valore sarà 0 per RAD1 e $\pi \times 2$ per RAD2; se vengono omessi entrambi si avrà una circonferenza completa.

Esempio: per tracciare una semicirconferenza di centro 200,0 e raggio 100 scrivere:

ARCO 200,0,100,0,3.1415

Per tracciare un cerchio di centro 220,100 e di raggio 200 scrivere:

ARCO 220,100,200

Per ottenere ellissi o parti di ellissi è necessario usare il comando SCALA che verrà descritto più avanti.

SETTORE

Il comando SETTORE ha la stessa sintassi del comando ARCO, ma differisce da

esso in quanto, oltre a tracciare la parte di circonferenza, unisce gli estremi di quest'ultima con il centro, ottenendo un settore di circonferenza.

ZERO

ZERO porta la penna all'origine dell'attuale sistema di riferimento, ossia alle coordinate 0,0. Non accetta parametri.

REL

REL imposta il sistema di riferimento definito dall'utente, traslando gli assi alle coordinate stabilite con il comando **ZERO=**. Non accetta parametri.

ASS

ASS riporta al sistema di riferimento normale, ossia a quello attivato all'accensione, con origine 0,0 all'estrema sinistra.

ZERO=

ZERO= serve a stabilire la nuova origine relativa su cui si baseranno tutti i prossimi comandi grafici. Perchè sia attiva deve essere accompagnata dal modo relativo, attivato dal comando **REL**.

Deve essere seguito dalle coordinate del punto in cui si vuol fissare la nuova origine. Naturalmente le coordinate di questo punto sono riferite al sistema assoluto, con origine all'estrema sinistra.

Esempio:

ZERO= 220,0

REL

La nuova origine è 220,0.

Tutti i prossimi comandi sono ora riferiti a 220,0.

Esempio: dopo aver digitato i due comandi **ZERO=** e **REL**, digitare: **LINE 0,0 to 100,100**; si noterà che la linea tracciata ha in realtà coordinate $0+200,0+0$ e $220+100,0+100$ ossia 220,0 e 320, 100.

SCALA

Il comando **SCALA**, il cui scopo è di poter cambiare le proporzioni alle figure che vengono stampate dal 1520, fissa quattro parametri:

SCALA Xper,Yper,Xadd,Yadd

Xper è la quantità per la quale vengono moltiplicate tutte le coordinate **X** inviate al plotter. Per avere dei disegni corrispondenti alle coordinate assegnate al basic, è necessario che questo parametro sia impostato a 1, che è il suo valore all'accen-

sione. Yper è la quantità per la quale vengono moltiplicate tutte le coordinate Y. Da notare che è possibile rovesciare delle figure semplicemente impostando con una quantità negativa uno o tutti e due i parametri di moltiplicazione; è possibile allungare, schiacciare, ridurre o ingrandire qualsiasi figura.

Xadd è la quantità di traslazione lungo l'ascissa; tutte le coordinate X, dopo essere state moltiplicate per Xper, vengono sommate ad Xadd. Lo stesso vale per Yadd, che è riferito alle coordinate Y.

Per conoscere i valori attuali di tali variabili, è necessario usare la funzione di SFA-SAMENTO (X), che ritorna il valore di: Xper quando X=0 : Yper quando X=1 : Xadd quando X=2 e Yadd quando X=3.

IMPORTANTE: Per resettare il plotter, la prima linea di ogni programma deve contenere i comandi SCALA e ASS, oppure SCALA e REL.

Per provare il comando SCALA, caricare il programma AQUILA dal dischetto del "Basotter" scrivendo: (vedi nota)

LOAD "AQUILA" (naturalmente dopo aver caricato "Basotter")

scrivete poi LIST

```
0 SCALA 1.4,1.2,0,0
5 PENNA ROSSA
20 READ X,Y
30 VAI X,Y
40 FOR I = 0 TO 62 : READ X,Y : LINEA TO
  X,Y : NEXT
50 READ X,Y : VAI X,Y
60 FOR I = 0 TO 2 : READ X,Y : LINEA TO
  X,Y : NEXT : ZERO : CLR
1000 DATA10,20,330,20,338,27,340,35,215,
  35,215,40,330,40,338,47,340,55
1010 DATA222,55,222,60,330,60,338,67,340
  ,75,230,75,230,80,330,80,338,87
1020 DATA340,95,230,95,223,92,195,40,190
  ,35,182,35,175,40,172,50,170,70,171,77
1030 DATA175,82,180,84,185,84,193,82,192
  ,92,185,95,155,95,147,92,128,50,120,40
1040 DATA112,32,102,34,101,43,102,55,115
  ,75,121,84,120,91,118,95,0,95,2,88
1050 DATA10,80,90,80,90,75,0,75,2,68,10,
  60,90,60,90,55,0,55,2,48,10,40,90,40
1060 DATA90,35,0,35,2,28,10,20
1070 DATA165,91,175,91,170,88,165,91
```

Modificate la linea 0, andandoci sopra con il cursore, e scrivete:

0 SCALA 0.5,-1,100,0

poi scrivete RUN e osservate ciò che accade.

L'aquila si è ristretta e nello stesso tempo si è anche rovesciata!

Provate a scrivere:

PRINT SFASAMENTO (0),SFASAMENTO (1),SFASAMENTO (2),SFASAMENTO (3)

Il risultato deve essere:

.5 -1 100 0

Per capire meglio come funziona questo comando, fate un po' di esperimenti cambiando i parametri nella linea 0.

NOTA: I comandi LOAD e SAVE assumono come device di default il numero 8 (floppy) e come secondary address il numero 1; per caricare un programma da cassetta occorrerà quindi specificare il numero di device 1 (LOAD "Nome del programma", 1).

COMANDI DI GESTIONE STRINGHE

Il 1520 oltre a disegnare linee, cerchi, settori ecc. può anche scrivere con quattro diversi tipi di caratteri, maiuscoli e minuscoli, sia in modo normale che inclinati di 90 gradi a destra.

I comandi atti a gestire la stringhe sono i seguenti:

Versione italiana

MISURA
PLIST
STAMPA
INCLINAZIONE
MAIUSCOLE
MINUSCOLE

Versione inglese

SIZE
PLIST
PPRINT
SLOPE
UPPER
LOWER

STAMPA

Il comando STAMPA serve a stampare parole, frasi o titoli con il 1520. Deve essere seguito o da una frase racchiusa tra virgolette, o da una variabile stringa o da una funzione stringa.

Esempio: per scrivere 'CARO AMICO TI SCRIVO...'

STAMPA "CARO AMICO TI SCRIVO"

oppure:

A\$="CARO AMICO TI SCRIVO":STAMPA A\$

Per stampare dei numeri è necessario ricorrere alla funzione STR\$

Esempio: per scrivere una serie di 20 numeri iniziando da 1 e finendo a 20, scrivere il seguente programmino:

```
10 FOR I = 1 TO 20
```

```
20 STAMPA STR$(I) : NEXT
```

poi scrivere RUN

Il 1520 stamperà uno dopo l'altro i numeri dall'uno al venti.

È da notare che il comando stampa, quando non è seguito da parametri, invia al plotter un ritorno carrello, e fissa a capo la nuova origine assoluta.

MISURA

Il comando MISURA stabilisce il tipo dei caratteri con il quale si vuole scrivere. Deve essere seguito da una quantità compresa tra 0 e 3. 0 indica che si vuole scrivere con i caratteri più piccoli, 3 con i più grossi.

Scrivendo con i caratteri '0' si hanno a disposizione ben 80 colonne, ed il 1520 può essere comodamente utilizzato anche come stampante con i programmi di word processing, quali Easy script, con '1' si hanno 40 colonne, con '2' 20 e con '3' 10.

Provate a scrivere:

MISURA 0:STAMPA "Basotter 1984"

MISURA 3:STAMPA "Basotter 1984"

Il primo comando imposta i caratteri ad 80 colonne, ed il 1520 scriverà "Basotter 1984" con dei caratteri piccolissimi, il secondo imposta le 10 colonne, ed il risultato è in caratteri enormi, tanto che "Basotter 1984" non sta su una sola linea, e 84 viene scritto a capo.

PLIST

PLIST agisce come LIST ma invece di visualizzare un programma sullo schermo lo scrive sul plotter. Per il funzionamento del comando PLIST leggere il comando LIST sul manuale del Commodore 64.

INCLINAZIONE

INCLINAZIONE fissa il modo scrittura con caratteri normali o ruotati di novanta gradi. Deve essere seguito da una quantità compresa tra 0 e 1. 0 per caratteri dritti, 1 per caratteri ruotati.

Esempio:

INCLINAZIONE 1 : STAMPA "CIAO"

INCLINAZIONE 0 : STAMPA "CIAO"

il primo comando stampa CIAO con caratteri ruotati, il secondo con caratteri normali.

MAIUSCOLE-MINUSCOLE

MAIUSCOLE seleziona i caratteri maiuscoli, mentre MINUSCOLE quelli minuscoli.

Esempio:

MAIUSCOLE:STAMPA "CARATTERI MAIUSCOLI":MINUSCOLE:STAMPA
"CARATTERI MINUSCOLI"

I COMANDI "BASIC"

Questi comandi facilitano la programmazione basic, rendendo più agevole il passaggio sul Commodore 64 di programmi creati per altri computers, e rendendo più potente il basic del C64. I comandi sono i seguenti:

Versione italiana	Versione inglese
ELSE	ELSE
CGOTO	CGOTO
CGOSUB	CGOSUB
CPOKE	CPOKE
JOY	JOY
PADDLE	PADDLE
HEX\$	HEX\$
BIN\$	BIN\$
CPEEK	CPEEK

ELSE

ELSE riprogramma il comando IF THEN, che normalmente esegue i comandi che seguono THEN solo se la condizione che segue IF è vera, ossia se il valore o il risultato dell'operazione logica che segue IF è diverso da 0.

In questo modo è possibile utilizzare la struttura IF..THEN..ELSE.

Se il risultato dell'operazione che segue IF è vero allora vengono eseguiti i comandi che seguono THEN ed ignorati quelli che seguono ELSE, se è falso allora accade il contrario, cioè sono eseguiti i comandi che seguono ELSE ed ignorati quelli che seguono THEN.

Esempio:

```
10 INPUT A
20 IF A = 10 THEN STAMPA "A=10" : ELSE STAMPA "A<> 10"
30 GOTO 10
RUN
```

il programma chiede di scrivere un valore; se questo è 10 allora scrive "A=10" sul 1520, altrimenti scrive "A <> 10".

CGOTO-CGOSUB

CGOTO e CGOSUB hanno la stessa funzione dei comandi basic GOTO e GOSUB, però accettano anche formule e variabili, senza dover ricorrere all'istruzione ON GOTO/GOSUB.

CPOKE

CPOKE è simile alla POKE del basic 64, ma tratta valori a 16 bit (cioè numeri compresi tra 0 e 65535), convertendoli in parte alta e parte bassa, mettendo la parte alta nel byte di indirizzo uguale al primo parametro dell'istruzione e la parte alta nel byte seguente.

Esempio:

```
CPOKE 10000,32768
```

mette nel byte 10000 il valore 0 e nel byte 10001 il valore 128.

CPEEK

La funzione CPEEK è come la funzione basic 64 PEEK, ma agisce ritornando un valore a 16 bit, con un procedimento simile a quello del comando CPOKE.

Esempio:

```
CPOKE 10000,23452
```

```
PRINT CPEEK (10000)
```

23452 viene visualizzato sullo schermo.

JOY

La funzione JOY serve a leggere lo stato di uno dei due joystick applicabili al Commodore 64. È necessario specificare quale dei due joystick si vuole leggere; per leggere il joystick nella control port 1 il valore corrispondente è 1, per leggere quello nella control port 2 il valore da specificare è 0.

Esempio:

```
10 PRINT JOY (0),JOY (1):GOTO10
```

```
RUN
```

Si vedrà una doppia fila di numeri scorrere sullo schermo; quella di sinistra corrisponde al joystick in porta 2, quella di destra al joystick in porta 1.

I valori riportati dalla funzione sono i seguenti:

- 1: joy in alto
- 2: joy in basso
- 4: joy a sinistra
- 8: joy a destra

128: fire

Se il joystick è posto in diagonale, il valore assunto dalla funzione JOY sarà la somma dei due valori corrispondenti.

Esempio: in alto-sinistra = $1+4 = 5$

joy in basso-destra-fire = $2+8+128=138$

PADDLE

La funzione PADDLE legge la posizione del potenziometro di una delle quattro paddle. È necessario specificare tra parentesi il numero della paddle che si vuole leggere, che deve essere compreso tra 0 e 3. Per rivelare il fire delle paddle si deve ricorrere alla funzione JOY.

HEX\$

La funzione HEX\$ trasforma una quantità numerica decimale in formato stringa esadecimale. Se la quantità decimale è minore di 256 allora la stringa esadecimale sarà lunga 2 caratteri, altrimenti sarà lunga 4.

Esempio:

```
PRINT HEX$ (240)
```

il 64 scriverà F0.

```
PRINT HEX$ (65535)
```

il 64 scriverà FFFF.

BIN\$

La funzione BIN\$ agisce in modo simile a HEX\$, solo che riporta una stringa binaria anziché esadecimale. La stringa sarà lunga 8 caratteri se la quantità da convertire è minore di 256, altrimenti sarà lunga 16 caratteri.

Esempio:

```
PRINT BIN$ (64)
```

RISPOSTA: 01000000

```
PRINT BIN$
```

RISPOSTA: 1111111111111111

I COMANDI DI GESTIONE DISCO

Basotter aggiunge anche dei comandi che servono a gestire le funzioni del drive in modo più semplice e veloce, rendendo possibili le comunicazioni senza dover continuamente aprire e chiudere noiosissimi canali di comunicazione. I comandi disco sono i seguenti:

Versione italiana

ELENCA

CANCELLA

CONVALIDA

RINOMINA

DISTRUGGI

INIZIALIZZA

DISCO

]

Versione inglese

DIRECTORY

SCRATCH

VDATE

RENAME

FMAT

INITIALIZE

DISK

]

ELENCA

Questo comando visualizza sullo schermo la directory del disco, ossia l'elenco dei programmi/files presenti sul dischetto nel drive. Non necessita di parametri, e visualizza tutti i files e programmi. Al posto di ELENCA può essere usato il comando] che è sicuramente più veloce e comodo, e consente inoltre di visualizzare la directory con selezione. Vedere comando].

CANCELLA

Questo comando serve a cancellare un file o un programma o una serie di files/programmi dal dischetto presente nel drive. Deve essere seguito da un parametro stringa, il cui contenuto è un nome o una serie di nomi separati da virgola; essi devono corrispondere ai nomi dei files/programmi da cancellare.

Esempio:

1: CANCELLA "PROG1"

2: CANCELLA "PROG1,PROG2,PROG3"

3: CANCELLA "PRO*"

L'esempio 1 cancella il programma o file che ha per nome 'prog1'. L'esempio 2 cancella i programmi/files prog1, prog2 e prog3, l'esempio 3 cancella tutti i files/programmi il cui nome comincia con pro. per usare le 'pattern', ossia i metodi per cancellare più files con un comando solo, leggere il manuale del drive.

CONVALIDA

CONVALIDA esegue il comando 'VALIDATE', descritto nel manuale del drive. Non deve essere seguito da parametri. La sua funzione è quella di ricontrollare la BAM (Block Available Map, ossia mappa dei blocchi liberi), eventualmente modificandola qualora non corrisponda alla reale memoria occupata dai files/programmi.

RINOMINA

Questo comando cambia il nome di un file o programma. Deve essere seguito da una variabile o costante stringa.

Esempio: per cambiare il nome di PROG1 in PROGRAMMA, scrivere:

RINOMINA "PROGRAMMA=PROG1"

Il primo nome è il nuovo nome che si vuol dare al programma; esso deve essere seguito da un =, poi vi è il vecchio nome che si vuole cambiare.

Naturalmente il file o programma non viene assolutamente modificato.

DISTRUGGI

Questo comando è l'equivalente del comando 'NEW' descritto nel manuale del drive, e serve a formattare un dischetto. L'operazione è necessaria qualora si abbia un dischetto nuovo, che deve essere inizializzato prima di poter essere usato normalmente. Tuttavia può venir formattato anche un disco già usato, in tal caso tutti i dati presenti su di esso verranno persi.

È necessario attribuire un nome ed un identificatore al disco che si vuole formattare; tuttavia se il disco è già stato usato con il drive del 64 è possibile omettere l'identificatore; ciò comporta un notevole risparmio di tempo ma il drive non cancellerà fisicamente tutti i dati presenti sul disco, di cui verrà azzerata solamente la directory.

Esempio: per formattare un dischetto con nome '-COMMODORE 64-' ed identificatore '64' scrivere:

DISTRUGGI "-COMMODORE 64-,64"

Per cancellare un dischetto già usato è possibile scrivere:

DISTRUGGI "-COMMODORE 64-"

In tal caso il disco manterrà l'identificatore precedente.

INIZIALIZZA

Questo comando serve ad inizializzare il disco; è l'equivalente del comando INITIALIZE descritto nel manuale del drive. Non necessita di alcun parametro.

DISCO

Il comando DISCO serve per leggere lo stato del disco e per comunicare qualsiasi comando al drive. Se non è seguito da parametri legge e visualizza lo stato, altrimenti comunica al drive i parametri, che devono essere stringhe.

Esempio: DISCO

Se il disco non è in errore, verrà visualizzato: 00,OK,00,00

DISCO "11"

Inizializza il drive 1 (se si ha una unità a doppio drive 4040 o 3040), il che non è possibile con il comando INIZIALIZZA, che inizializza sempre il drive 0.

COMANDO "J"

La parentesi quadra destra ha una doppia applicazione; se usata senza parametri visualizza lo stato del disco, nella stessa maniera del comando DISCO, altrimenti deve essere seguita da un '\$', e in tal caso visualizzerà la directory, come il comando ELENCA. Ciò che è possibile in più rispetto a quest'ultimo comando, è la visualizzazione della directory con selezione.

Esempio:

]\$ directory normale

]\$:pro visualizza solo i files/programmi che iniziano con pro.

COMANDI BASIC "BASOTTER 84"

penna	nera/blu/verde/rossa cambia il colore della penna del plotter
zero= x,y	porta la penna alle coordinate x,y e setta l'origine relativa
zero	porta la penna all'origine relativa
linea: to x,y to x1,y1...	
linea: x,y to x1,y1...	congiunge i punti x,y; x1,y1...
rel	setta i parametri di output in scala relativa all'origine settata con zero=
vai x,y	porta la penna alle coordinate x,y
ass	setta i parametri di output in scala assoluta: tutte le coordinate si riferiranno all'origine assoluta (0,0)
trat x	selezione il tratteggio: 0=linee continue: 1-15 = tipo di tratteggio
arco x1,y1,r,rad1,rad2	traccia l'arco di centro x1,y1; di raggio r nell'intervallo compreso tra rad1 e rad2 (in radianti)
sector x1,y1,r,rad1,rad2	come arco, ma congiunge l'arco con il centro. (i parametri rad1 e rad2 possono essere omessi, in tal caso il loro valore sarà 0 per rad1 e 2 pi per rad2)
rettangolo x,y to x1,y1	disegna il rettangolo di diagonale x,y to x1,y1
misura x	(x(4) seleziona i caratteri di dimensioni x
plst	lista il programma in memoria sul plotter
stampa	come print ma su plotter
inclinazione x	x=0 o x=1 0=caratteri normali; 1=caratteri ruotati in senso orario di 90 gradi
scala x,y,k,h	x=fattore di moltiplicazione delle coordinate x di tutti i comandi; y fattore di moltiplicazione y; k = fattore di traslazione x; h =fattore di traslazione y
maiuscole	seleziona i caratteri maiuscoli
minuscole	seleziona i caratteri minuscoli
elena	visualizza la lista dei file/programmi presenti sul disco
cancella	cancella sul disco un file/programma
convalida	esegue la validate del disco
rinomina	cambia il nome di un file/prg su disco
distruggi	formatta il disco, assegnandogli il nome dato

inizializza
disco
disco "com"
cgoto
cgosub
else
italiano
english
cpoke

FUNZIONI

Joy (x)
paddle (x)
hes\$ (x)
bin\$ (x)
sfasamento (x)

cpeek (x)

inizializza il drive
visualizza lo stato del disco
fa eseguire il comando (com) al drive
come goto ma accetta formule e variabili
come gosub...
struttura if...then...else
cambia le Keywords da inglese a italiano
cambia le Keywords da italiano a inglese
come poke ma a 16 bits

ritorna la posizione del joystick x
ritorna il valore della paddle x
ritorna la stringa esadecimale corrispondente a x
come hex\$ ma binaria
x=1: ritorna il fattore x del comando scale; x=2
il fattore y; x=3 k; x=4 h
ritorna il valore a 16 bit della locazione
x+256*locazione x+1.a

AQUILA

```

0 SCALA 1.4,1.2,0,0
5 PENNA ROSSA
20 READ X,Y
30 VAI X,Y
40 FOR I = 0 TO 62 : READ X,Y : LINEA TO
  X,Y : NEXT
50 READ X,Y : VAI X,Y
60 FOR I = 0 TO 2 : READ X,Y : LINEA TO
  X,Y : NEXT : ZERO : CLR
1000 DATA10,20,330,20,338,27,340,35,215,
35,215,40,330,40,338,47,340,55
1010 DATA222,55,222,60,330,60,338,67,340
,75,230,75,230,80,330,80,338,87
1020 DATA340,95,230,95,223,92,195,40,190
,35,182,35,175,40,172,50,170,70,171,77
1030 DATA175,82,180,84,185,84,193,82,192
,92,185,95,155,95,147,92,128,50,120,40
1040 DATA112,32,102,34,101,43,102,55,115
,75,121,84,120,91,118,95,0,95,2,88
1050 DATA10,80,90,80,90,75,0,75,2,68,10,
60,90,60,90,55,0,55,2,48,10,40,90,40
1060 DATA90,35,0,35,2,28,10,20
1070 DATA165,91,175,91,170,88,165,91

```

DIAGRAMMI

```

10 SCALA 1,1,0,0 : PENNA BLU : ASS : TRA
T 0
20 MISURA1 : STAMPA : STAMPA
30 FORI=0TO5:A(I)=RND(1)*10:NEXT
40 STAMPA"  -- TABELLA DATI --" : STAMP
A : PENNA NERA
50 VAI 0,-40:STAMPA
60 FORI=0TO5:STAMPA"DATO"+STR$(I+1)+": "
: VAI200,0:STAMPASTR$(A(I)):STAMPA:NEXT
65 PENNA BLU : STAMPA:STAMPA:STAMPA"VISU
ALIZZAZIONE DIAGRAMMA A BARRE"
70 STAMPA:STAMPA:LINEA 60,0 TO 60,-200 T
O 450,-200 : GOSUB200
71 VAI0,-200:PENNA VERDE
72 MISURA0 : FORI=0TO5 :VAI 70+60*I,-230
:STAMPA"DATO:"+STR$(I+1):NEXT:MISURA1
80 FORI=0TO10 : VAI 0,-200+I*20 : TRAT 0
: STAMPARIGHT$(STR$(I),2)
85 TRAT 7 : LINEA TO 60,-200+I*20
90 NEXT
100 PENNA ROSSA : TRAT0 : FORI=0TO5
110 RETTANGOLO 70+60*I,-199 TO 110+60*I,
-199+A(I)*20 : GOSUB300 :NEXT
119 STAMPA:STAMPA:STAMPA:STAMPA:STAMPA
120 STAMPA : STAMPA : STAMPA"VISUALIZZAZ
IONE DIAGRAMMA A TORTA"
140 ZERO= 220,-170 : J=0 : REL
150 FORI=0TO5: J=J+A(I):NEXT:TP=(2*PI)/J: J
=0
160 FORI=1TO6:R(I)=J+A(I-1)*TP/2:T(I)=J+
A(I-1)*TP:J=T(I):NEXT
165 FORI=1TO6:PENNA (IAND3)
170 SETTORECOS(R(I))*10,SIN(R(I))*10,100
,T(I-1),T(I):NEXT:MISURA0
175 PENNA NERA : ASS
180 FORI=1TO6:VAI220+COS(R(I))*150,-170+
SIN(R(I))*150
190 STAMPALEFT$(STR$(A(I-1)),5):NEXT:FOR
I=0TO10:STAMPA:NEXT:END

```

```
200 LINEA 60,0 TO 65,5 TO 65,-195 TO 60,  
-200:LINEA65,-195TO70,-195  
210 FORI=1TO6:LINEA55+I*60,-195TO70+I*60  
,-195 :NEXT  
220 RETURN  
300 LINEA70+60*I,-199+A(I)*20TO75+60*I,-  
194+A(I)*20TO115+60*I,-194+A(I)*20  
310 LINEA TO110+60*I,-199+A(I)*20  
320 LINEA115+60*I,-194+A(I)*20TO115+60*I  
,-194TO110+60*I,-199:RETURN
```

SIN X + SIN X/2

```

10 ASS : TRAT0 : INCLINAZIONE0 : SCALA1,
1,0,0
20 PENNA ROSSA : VAI 50,0 : MISURA 2
30 STAMPA"Y" : PENNA BLU : STAMPA"=SIN X
+SIN X/2": VAI 0,-400 : STAMPA : PENNA
NERA
35 ZERO= 240,0 : REL : RETTANGOLO -240,4
00 TO 239,-400
40 LINEA 0,400 TO 0,-400 : LINEA -240,0
TO 240,0 : VAI 0,-400 : PENNA BLU
50 FOR X=-400 TO 400 STEP 3
60 LINEA TO -SIN(X/25)*100,X : NEXT
70 VAI0,400 : FOR X=-400 TO 400 STEP 10
80 PENNA VERDE : VAI 0,-400
90 FOR X=-400 TO 400 STEP 3
100 LINEA TO -SIN(X/50)*100,X : NEXT
110 PENNA ROSSA : VAI 0,-400
120 `FOR X=-400 TO 400 STEP 3
130 LINEA TO -SIN(X/50)*100-SIN(X/25)*10
0,X : NEXT
140 VAI 0,-400
150 MISURA1 : PENNA BLU : PLIST

```




 **commodore**
COMPUTER